

---

Subject: Wealth Index - Uganda 2000

Posted by [fewerling](#) on Thu, 17 Nov 2016 12:38:34 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

Dear all,

We are trying to reproduce the results for skilled birth attendance (doctor, nurse/midwife, clinical officer, private nursing aid) by wealth quintiles in Uganda 2000 (Table 9.7).

However, using v190, the wealth quintiles provided with the survey, the coverage distribution does not match the published estimate.

Using the doctor attendance as example, the published estimates are:

```
Wealth index quintile
Lowest          1.6
Lower middle    1.9
Middle          2.1
Upper middle    3.3
Highest         12.4
```

Using the sample weights and the codes below, we found:

```
. svyset v021 [pweight=v005]
. svy:prop m3a, over(v190)
```

```
Wealth index quintile
Lowest          3.7
Lower middle    9.2
Middle          2.6
Upper middle    2.6
Highest         3.0
```

The national estimate matches the published. Our only problem is with the wealth index. We also tried to re-calculate the wealth quintiles using the continuous wealth scores (v191):

```
xtile wealth=v191 [pweight=v005], nq(5)
ta wealth v190
```

```
5 |
quantiles |          wealth index quintile
of v191 | lowest 20 next to l middle 20 next to h highest 2 | Total
-----+-----+-----+-----+-----+-----+-----+-----+-----+-----+
1 | 1,310      0      0      0      0 | 1,310
2 | 53      1,034      503      0      0 | 1,590
3 | 0      0      758      708      0 | 1,466
4 | 0      0      0      636      715 | 1,351
5 | 0      0      0      0      1,396 | 1,396
-----+-----+-----+-----+-----+-----+
Total | 1,363      1,034      1,261      1,344      2,111 | 7,113
```

The quintiles provided do not match the calculated ones, however, with the calculated quintiles we still could not reproduce the report results.

We are using the dataset version 41. Is it possible that the results were calculated in the older version, and that the updated one have a problem with the provided wealth index?  
If not, how should we proceed?

Thank you in advance!

---

Subject: Re: Wealth Index - Uganda 2000  
Posted by [Trevor-DHS](#) on Wed, 23 Nov 2016 23:23:18 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

I'm wondering if you have a problem with the way v190 is merged into your dataset. The copy of the dataset I have here does not include v190, but instead it needs to be merged from a file of wealth quintiles. Here is my code:

```
* Merge the wealth index into the dataset
use "C:\Data\DHS_Stata\UGKR41FL.DTA", clear
gen whhid = substr(caseid,1,12)
merge m:1 whhid using "C:\Data\DHS_Stata\UGWI41FL.dta"
keep if _merge==3
clonevar v190=w1thind5
```

```
* generate a weight variable and tabulate as in the table
gen wt=v005/1000000
```

```
* treat the few missing cases as No for doctor
replace m3a = 0 if m3a == .
```

```
* tabulate
tab v190 m3a [iw=wt],m row
```

```
* now using svy:prop, taking stratification into account
egen stratum=group(v024 v025)
```

```
svyset v021 [pweight=wt], strata(stratum)
svy:prop m3a, over(v190)
```

The tab and svy:prop command produce results that match the table in the report.

---

Subject: Re: Wealth Index - Uganda 2000  
Posted by [fewerling](#) on Mon, 05 Dec 2016 14:16:34 GMT  
[View Forum Message](#) <> [Reply to Message](#)

---

Thank you, Trevor!

The problem was solved.

Best,  
Fernanda

---